
' PURPOSE
' The xMacroMorsels are intended to be for Excel plus

' Xnumbers what the MacroMorsels are for Excel plus VBA:

' miniature macros that illustrate particular aspects of

' Excel and Xnumbers.ddl that, otherwise, might easily get

' you stuck. They are stand-alone macros that can be used

' to illustrate as well as to test the highlighted aspect.

' For all other introductory remarks, see the MacroMorsels.

' The reason to place these xMacroMorsels in a separate

' collection is that they require that Xnumbers.ddl be

' installed on your computer. If they were included with

' the regular MacroMorsels, and were then used on computers

' on which Xnumbers had not been installed, they would

' produce endless compile errors!

' This collection is a work in progress, and here you see

' only its very first installment. Your suggestions and

' comments are invited. Please send them by email to

' rdelevie@bowdoin.edu

' COPYRIGHT and ABSENCE OF WARRANTY
' These xMacroMorsels are copyrighted, and are licenced

' under the GNU General Public License. In publications

' that use them, please acknowledge this web site as

' their source, so that other potential users can also

' learn about (and possibly benefit from) them.

' These xMacroMorsels are made available without any

' warranty whatsoever. Prospective users are advised to

' check the performance of these xMacroMorsels on their

' own computer to make sure that they work properly in

' this setting. Some operations may not work in early

' versions of Excel; some features used here may not be

' available in your particular version of Excel. Some

' conventions and commands may be different in non-US

' versions. And, of course, these xMacroMorsels will not

' work if Xnumbers.dll is not installed on your computer.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'                                                         '

'                     THE xMACROMORSELS                   '

'                                                         '

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ExtractFullPrecisionFromSpreadsheet()

' When you try to read a number from the spreadsheet into

' VBA, all you care about is the 15 digits that the spread-

' sheet can maximally display, because VBA cannot handle any

' larger numberlengths either. But when you extract a number

' from the spreadsheet into an Xnumbers macro, it may be

' useful to get also their normally hidden tails, i.e., the

' parts rounded just before the screen display. Here is how

' you can do that. But keep in mind: the gain is minor: the

' relative precision epsilon is about 2E-16, the relative 

' precision of the spreadsheet display between 1E-15 and 

' 1E-14, depending on the magnitude of the leading decimal 

' digit. Moreover, these are not the actual, binary numbers, 

' but their decimal equivalents.

' The trick is to extract the spreadsheet number first,

' round it to 15 decimals, then extract the difference, and

' finally add the rounded value back. This may sound

' complicated, but it works.

' Here is an example. In a spreadsheet cell place an

' instruction such as = 7 + 5E-16, i.e., the sum of two

' numbers x and y where x >> y and abs(y) < abs(x) * 1e-15

' (so that y will seem to have no effect on the sum) but

' also abs(y) > 2^-53 * abs(x) or approximately

' abs(y) > abs(x) * 1E-16. Note that this is a very narrow

' (just one decade wide) constraint on y.

' Make sure the Immediate Window is open. Then call this

' xMacroMorsel. The print.debug statements illustrate what

' each line does.

Dim MP As Xnumbers

Set MP = New Xnumbers

With MP

Dim B As String, i As Integer

Dim vW, vX, vY, vZ

vW = Selection.Value

  Debug.Print "vW = " & vW

vX = .xRoundr(vW, 15)

  Debug.Print "vX = " & vX

vY = (Selection.Value - vX)

  Debug.Print "vY = " & vY

vZ = .xAdd(vY, vW)

  Debug.Print "vZ = " & vZ

  Debug.Print ""

End With

Set MP = Nothing

End Sub

''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub DisplayFullPrecisionResultOnSpreadsheet()

' Upon writing Xnumbers results back onto the spreadsheet

' they are converted to 15 figures. While that is usually

' precisely what we want, there may be circumstances where

' it would be useful to display the unabridged Xnumbers

' result on the spreadsheet. (You can always display them

' unobtrusively in the Immediate Window, but when others

' will also use your macro, that may not be desirable.)

' With Xnumbers.ddl this cannot be done as a number, but is

' possible as a string. Here is a convenient ways to do so.

' Highlight an empty cell, then call the macro. (If the

' cell is not empty, its contents will be overwritten.)

Dim MP As Xnumbers

Set MP = New Xnumbers

With MP

.DigitsMax = 50

Dim a As String

a = "1.23456789012345678901234567890"

Selection = Chr(39) & a

' Please note that you cannot use this result directly for

' subsequent calculations because, as you can see in the

' formula box, the apostrophe (the character #39) added as

' a prefix, makes it a text string rather than a number.

' As soon as you remove this apostrophy, or copy the cell

' content without that apostrophe into another cell, the

' number will be truncated to 15 figures.

' You can also display the number with only up to a fixed

' number of decimals (in this example: 25), by first

' rounding the number to the required precision. To do so,

' just re-activate the following two lines, of which the

' first merely prevents overwriting the earlier output:

'Selection.Offset(1, 0).Select

'Selection = Chr(39) & .xRoundr(a, 25)

' Of course, you get the regular output once you reactivate

'Selection.Offset(1, 0).Select

'Selection = a

' A less obtrusive way is to display the number in a cell

' comment, again as a string. Here we will combine the regular

' output with an attached cell comment showing its full

' precision. Reactivate the next lines:

'ActiveCell.ClearComments

'ActiveCell.AddComment

'ActiveCell.Comment.Visible = False

'ActiveCell.Comment.Shape.Width = 150

'ActiveCell.Comment.Shape.Height = 40

'ActiveCell.Comment.Text Text:="The full precision result is" _

  & Chr(10) & a

' If you do not need all these extra digits, here is a useful,

' compact alternative that also allows you to anticipate how

' wide the comment box needs to be:

'Selection.Offset(2, 0).Select

'a = "2.24466880011335577991133557799"

'Selection = a

'a = .xRoundr(a, 25)

'ActiveCell.ClearComments

'ActiveCell.AddComment

'ActiveCell.Comment.Visible = False

'ActiveCell.Comment.Shape.Width = 136

'ActiveCell.Comment.Shape.Height = 12

'ActiveCell.Comment.Text Text:="a = " & a

End With

Set MP = Nothing

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ReconstituteAnEquationInXnumbers1()

' This xMacroMorsel illustrates how to reconstitute a

' spreadsheet expression F(x) of a single argument x in

' Xnumbers, and how to evaluate it with MP.xEval. The

' argument x can occur multiple times in the function F(x).

' Because MP.xEval does not use Application., code must be

' used to correct (at least) the most prevalent conflicts

' between Excel and VBA.

' Place a numerical value in a cell, and a function using

' that value in the cell to its right. Highlight both

' cells, then call this xMacroMorsel. All intermediary and

' final results are displayed in the Immediate Window.

' In order to validate this method, place some number

' in, e.g., cell B2, and in cell C2 an equation such as

' =3*B2^3+10+2*LOG(B2)-SQRT(B2) or =EXP(B25)*SIN(B25)

' that tests its operation. Highlight cells B2:C2, then

' call the MacroMorsel.

Dim MP As Xnumbers

Set MP = New Xnumbers

With MP

.DigitsMax = 50

'Dim vX As Double

Dim vX

Dim aX As String, CFormula As String

Dim fF As String, X As String

Dim vF

' Read the input

aX = Selection(1, 1).Address

vX = Selection(1, 1).Value

fF = Selection(1, 2).Formula

' Strip off the dollar signs in the address

aX = Replace(aX, "$", "")

  Debug.Print "aX = " & aX

  Debug.Print "vX = " & vX

  Debug.Print "fF = " & fF

' Reconstitute the equation for F(x) minus the equal sign

CFormula = Right(fF, Len(fF) - 1)

  Debug.Print "1: CFormula = " & CFormula

' Correct for the most prevalent discrepancies

' between Excel and VBA. Note that, in MP.xEval,

' Ln apparently does NOT need to be replaced by Log.

On Error Resume Next

CFormula = Replace(CFormula, "SQRT", "Sqr")

CFormula = Replace(CFormula, "LOG", _

  "(1/log(10))*Log")

  Debug.Print "2: CFormula = " & CFormula

' Replace the spreadsheet address by the symbolic value X

CFormula = Replace(CFormula, aX, "X")

  Debug.Print "3: CFormula = " & CFormula

' This is the desired result. To test it,

' either modify the existing value of vX, 

' or specify an entirely new value for X

X = .xAdd(vX, .xPow(10, .xNeg(20)))

'X = "1.23400000000000000001" ' this alternative also works

  Debug.Print "X = " & X

' Evaluate the formula with the new value of vX

' Note that vF = .xEval(Application.CFormula, X)

' is NOT supported!

vF = .xEval(CFormula, X)

  Debug.Print "vF = " & vF

  Debug.Print ""

End With

Set MP = Nothing

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ReconstituteAnEquationInXnumbers2()

' This xMacroMorsel illustrates how to reconstitute a

' spreadsheet expression F(x) of a vector x of multiple

' (here: 3) arguments x(i) in Xnumbers, and how to

' evaluate it with MP.xEval. Each of the arguments x(i)

' can occur multiple times in the function F(x).

' Because MP.xEval does not use Application., code must be

' used to correct (at least) the most prevalent conflicts

' between Excel and VBA.

' Place a numerical values of x in a contiguous row, and

' a function using that row vector in the cell to its right.

' Highlight their combination, then call this xMacroMorsel.

' All intermediary and final results are displayed in the

' Immediate Window.

' In order to validate this method, place some numbers in,

' e.g., cells B2:D2, and in cell E2 an equation such as

' =3*B2^3+10+2*LOG(C2)-SQRT(D2) or =B2*EXP(C2)*SIN(D2)

' or =5+4*B8^3*C8^2*D8-3*B8*C8^2*D8^3/(C8-B8) that tests its

' operation. Highlight cells B2:E2, then call the xMacroMorsel.

Dim MP As Xnumbers

Set MP = New Xnumbers

With MP

.DigitsMax = 50

Dim a1, a2, a3                      ' extend list as needed

Dim i As Integer, iMax As Integer

Dim CFormula As String

Dim fF As String

Dim aX, vX, vF, X

' Read the input

iMax = Selection.Columns.Count

ReDim aX(1 To iMax - 1)

ReDim vX(1 To iMax - 1)

ReDim X(1 To iMax - 1, 2)

For i = 1 To iMax - 1

  aX(i) = Selection(1, i).Address

  vX(i) = Selection(1, i).Value

Next i

fF = Selection(1, iMax).Formula

' Place all X-values in the required matrix format

X(1, 1) = "a1": X(1, 2) = vX(1)

X(2, 1) = "a2": X(2, 2) = vX(2)

X(3, 1) = "a3": X(3, 2) = vX(3)     ' extend list as needed

' Strip off the dollar signs in the addresses

For i = 1 To iMax - 1

  aX(i) = Replace(aX(i), "$", "")

Next i

' Reconstitute the equation for F(x) minus the equal sign

  Debug.Print "fF = " & fF

CFormula = Right(fF, Len(fF) - 1)

  Debug.Print "1: CFormula = " & CFormula

' Correct for the most prevalent discrepancies

' between Excel and VBA. Note that, in MP.xEval,

' Ln apparently does NOT need to be replaced by Log.

On Error Resume Next

CFormula = Replace(CFormula, "SQRT", "Sqr")

CFormula = Replace(CFormula, "LOG", _

  "(1/log(10))*Log")

' Replace the spreadsheet addresses by

' simple letter-number combinations

For i = 1 To iMax - 1

  CFormula = Replace(CFormula, aX(i), "a" & i)

Next i

  Debug.Print "2: CFormula = " & CFormula

' Insert the necessary blank spaces to make it

' compatible with the requirements of .xEval

CFormula = Replace(CFormula, "+", Chr(32) & "+" & Chr(32))

CFormula = Replace(CFormula, "-", Chr(32) & "-" & Chr(32))

CFormula = Replace(CFormula, "*", Chr(32) & "*" & Chr(32))

CFormula = Replace(CFormula, "/", Chr(32) & "/" & Chr(32))

CFormula = Replace(CFormula, "^", Chr(32) & "^" & Chr(32))

Debug.Print "3: CFormula = " & CFormula

  Debug.Print ""

Debug.Print "     with spreadsheet X-values:"

vF = .xEval(CFormula, X)

  Debug.Print "vF = " & vF

' Evaluate the formula with a new set of X-values

X(1, 2) = 9.87

X(2, 2) = 6.54

X(3, 2) = 3.21

Debug.Print "     with new X-values:"

vF = .xEval(CFormula, X)

  Debug.Print "vF = " & vF

  Debug.Print ""

End With

Set MP = Nothing

End Sub

